

18/02/2011
Cập nhật: 23/11/2012

NGUYỄN XUÂN THÀNH

LẬP MÔ HÌNH ĐỊNH GIÁ TRÁI PHIẾU

Bài giảng này hướng dẫn học viên xây dựng mô hình định giá và tính lợi suất đến khi đáo hạn của trái phiếu trên Excel. Bài viết cũng giải thích các nguyên lý tài chính đằng sau các công thức tính. Trái phiếu (bond) là một chứng khoán kỳ hạn từ một năm trở lên trong đó chứng nhận người vay nợ một khoản tiền được xác định cụ thể cùng với các điều khoản liên quan tới việc hoàn trả khoản tiền này và lãi trong tương lai.

Chúng ta bắt đầu bằng cách khởi động Excel với một tập tin được tạo mới theo mặc định. Đổi tên bảng tính thứ nhất “Sheet1” thành “Pricing” và bảng tính thứ hai “Sheet2” thành “YTM”. Bảng tính “Pricing” được dùng để định giá trái phiếu. Bảng tính “YTM” được dùng để tính lợi suất đến khi đáo hạn của trái phiếu.

Ta xem xét trái phiếu chính phủ mã số TP4A4804 với các thông tin căn bản dưới đây.

Tên	Trái phiếu chính phủ TP4A4804
Mệnh giá	100.000 VND
Lãi suất coupon	9%/năm
Trả lãi	1 lần một năm (vào ngày 18/11 hàng năm)
Ngày phát hành	18/11/2004
Ngày đáo hạn	18/11/2019

Định giá trái phiếu

Ở bảng tính “Pricing”, nhập tựa đề “ĐỊNH GIÁ TRÁI PHIẾU” vào ô B1.

Ta bắt đầu từ khung thông tin đầu vào bằng cách đánh máy “NHẬP DỮ LIỆU” vào ô B3. Nhập các tiêu đề: “Ngày thanh toán”, “Ngày đáo hạn”, “Lãi suất coupon hàng năm”, “Lợi suất đến khi đáo hạn”, “Giá trị hoàn trả nợ gốc (% mệnh giá)”, “Số lần trả lãi trong năm” và “Hệ đếm ngày” lần lượt vào các ô từ B4 đến B10.

Ta định giá trái phiếu vào ngày 16/2/2009. Hãy nhập ngày này vào ô C4.¹ Trái phiếu đáo hạn vào ngày 18/11/2019. Đây là ngày hoàn trả nợ gốc cuối cùng. Thông tin này được nhập vào ô C5.

¹ Vì có nhiều quy ước ngày, nên để không bị nhầm lẫn ta nhập hàm tính này vào ô C4 = DATE(year, month, day) = DATE(2009,2,19).

Bài giảng này do Nguyễn Xuân Thành, giảng viên Chính sách công, Chương trình Giảng dạy Kinh tế Fulbright biên soạn. Các bài giảng của Chương trình Giảng dạy Fulbright được sử dụng làm tài liệu giảng dạy trên lớp học, chứ không phải là để ủng hộ, phê bình hay dùng làm nguồn số liệu cho một tình huống chính sách cụ thể.

Lãi suất coupon hàng năm (coupon rate) là lãi suất mà tổ chức phát hành (bond issuer) cam kết chi trả theo định kỳ tính trên cơ sở một năm. TP4A4804 trả lãi coupon với lãi suất 9%/năm. Đây là thông tin nhập vào ô C6.

Để định giá được trái phiếu, ta phải biết lợi suất đến khi đáo hạn (yield to maturity - YTM) để dùng làm suất chiết khấu. YTM là suất sinh lợi nội tại (IRR) nếu mà nhà đầu tư được hưởng nếu mua trái phiếu bây giờ và giữ cho tới khi đáo hạn. Vào thời điểm tháng 2 năm 2009, lãi suất dài hạn trên thị trường là 9,8%. Nhà đầu tư trái phiếu sẽ yêu cầu suất sinh lợi 9,8%. Ta nhập thông tin này vào ô C7.

Mệnh giá trái phiếu (face value, par value) là nợ gốc của trái phiếu (principal) được quy ước là giá trị làm cơ sở để tính lãi coupon phải trả định kỳ. Các trái phiếu khác nhau có mệnh giá khác nhau. Để thống nhất, ta quy ước mệnh giá trái phiếu bằng 100, mặc dù trên thực tế TP4A4804 có mệnh giá 100.000 VNĐ. Theo quy ước này, giá trái phiếu tính được sẽ là tỷ lệ phần trăm của mệnh giá. Thông thường, trái phiếu sẽ trả nợ gốc một lần đến khi đáo hạn bằng 100% mệnh giá. Ta nhập con số 100 vào ô C8. Đối với những trái phiếu đặc biệt, nợ gốc hoàn trả có thể lớn hơn hay nhỏ hơn mệnh giá.

Trái phiếu có thể trả lãi 1 lần hay nhiều lần trong năm. Trái phiếu trên thị trường tài chính quốc tế thường trả lãi 2 lần 1 năm (trả lãi bán niên) và trái phiếu trong nước ở Việt Nam thường trả lãi 1 lần 1 năm (trả lãi thường niên). Vậy, đối với TP4A4804, ta nhập 1 vào ô C9.

Hệ đếm ngày quy ước cách đếm ngày mà các tổ chức và thị trường tài chính sử dụng để tính lãi. Hệ đếm ngày 30/360 quy ước một tháng có đúng 30 ngày và một năm có đúng 360 ngày.² Hệ đếm ngày actual/360 tính số ngày thực tế theo lịch, nhưng vẫn quy ước một năm có 360 ngày. Hệ đếm ngày actual/365 tính số ngày thực tế theo lịch và quy ước một năm có 365 ngày. Hệ đếm ngày actual/actual tính số ngày thực tế theo lịch và số ngày thực tế trong năm (tức là tính 366 ngày cho năm nhuận).

Ở ô C10, ta nhập giá trị 0 hay bỏ trống nếu sử dụng hệ đếm ngày 30/360 kiểu Mỹ, 1 nếu sử dụng actual/actual, 2 nếu là actual/360, 3 nếu là actual/365 và 4 nếu là 30/360 kiểu Âu. Ở Việt Nam, giao dịch trái phiếu được thực hiện theo quy ước hệ đếm ngày thực tế/số ngày thực tế trong năm (actual/actual). Ta nhập 1 vào ô C10.

Khung thông tin tiếp theo là kết quả định giá. Ta nhập tiêu đề “KẾT QUẢ” vào ô B12. Sau đó, nhập “Giá yết (% mệnh giá)”, “Số ngày kể từ lần trả lãi trước”, “Số ngày trong kỳ trả lãi”, “Lãi tích tụ”, và “Giá thanh toán (% mệnh giá)” lần lượt vào các ô B13 đến B17.

Ta sử dụng hàm PRICE trong Excel để định giá trái phiếu với định dạng:

PRICE(settlement,maturity,rate,yld,redemption,frequency,basis)

Các thông tin cần khai báo để định giá trái phiếu gồm có ngày thanh toán (settlement), ngày đáo hạn (maturity), lãi suất coupon (rate), lợi suất đến khi đáo hạn (yld), giá trị hoàn trả nợ gốc (redemption), số lần trả lãi trong năm (frequency) và hệ đếm ngày (basis).

Dưới đây là công thức nhập trong ô C13 để định giá trái phiếu:

² Excel còn phân biệt giữa hệ 30/360 kiểu Mỹ và 30/360 kiểu Âu. Đối với hệ 30/360 kiểu Mỹ, nếu ngày bắt đầu rơi vào ngày 31 của tháng thì sẽ được coi là ngày 30. Trong trường hợp ngày kết thúc rơi vào ngày 31 của một tháng thì nếu ngày bắt đầu là ngày 30 hay trước ngày 30 của một tháng, thì ngày kết thúc được coi là ngày 1 của tháng tiếp theo, còn nếu ngày bắt đầu rơi vào ngày 31 của một tháng, thì ngày kết thúc được coi là ngày 30 của tháng kết thúc đó. Đối với hệ 30/360 kiểu Âu, nếu ngày bắt đầu hay ngày kết thúc rơi vào ngày 31 của tháng, thì được chuyển thành ngày 30 của tháng đó. Hệ đếm ngày kiểu Mỹ phổ biến hơn hệ đếm ngày kiểu Âu.

=PRICE(C4,C5,C6,C7,C8,C9,C10)


Hình 1: Định giá trái phiếu

	B	C	D	E
1	ĐỊNH GIÁ TRÁI PHIẾU			
2				
3	NHẬP DỮ LIỆU			
4	Ngày thanh toán	16/02/2009	C4=DATE(2009,2,16)	
5	Ngày đáo hạn	18/11/2019	C5=DATE(2019,11,18)	
6	Lãi suất coupon hàng năm	9.00%		
7	Lợi suất đến khi đáo hạn	9.80%		
8	Giá trị hoàn trả nợ gốc (% mệnh giá)	100		
9	Số lần trả lãi trong năm	1		
10	Hệ đếm ngày	1		
11				
12	KẾT QUẢ			
13	Giá yết (% mệnh giá)	94.746	C13=PRICE(C4,C5,C6,C7,C8,C9,C10)	
14	Số ngày kể từ lần trả lãi trước	90	C14=COUPDAYBS(C4,C5,C9,C10)	
15	Số ngày trong kỳ trả lãi	365	C15=COUPDAYS(C4,C5,C9,C10)	
16	Lãi tích tụ	2.219	C16=(C14/C15)*C6*C8/C9	
17	Giá thanh toán (% mệnh giá)	96.965	C17=C13+C16	

Như minh họa trong Hình 1, giá trái phiếu tính theo công thức ở trên là 94,746. Giá này là giá niêm yết trên sàn giao dịch và được gọi là giá sạch (clean price) hay giá phẳng (flat price). Giá yết là giá không gộp lãi, nên không phải là giá thực tế thanh toán giữa người bán và người mua trái phiếu.³

Trong ví dụ của chúng ta, người mua trái phiếu được hưởng trọn vẹn khoản lãi kế tiếp (vào ngày 18/11/09). Tuy nhiên, trong 365 ngày của kỳ tính lãi từ 18/11/08 đến 18/11/09, người mua chỉ nắm giữ trái phiếu trong 275 ngày, nên chỉ xứng đáng được hưởng một phần của khoản lãi trong kỳ. Cũng trong kỳ tính lãi đó, người bán nắm giữ trái phiếu trong 90 ngày, nhưng không hề được hưởng khoản lãi nào. Hình 2 minh họa việc phân chia khoản lãi trong kỳ cho thời gian người bán trái phiếu được hưởng và thời gian người mua trái phiếu được hưởng.

Hình 2: Lãi tích tụ



³ Giá yết sẽ đúng là giá thanh toán nếu thời điểm định giá rơi đúng vào ngày trả lãi định kỳ.

Để tính giá thực tế thanh toán (invoice price) hay còn gọi là giá bẩn (dirty price), ta cộng thêm khoản lãi tích tụ kể từ ngày giao dịch cho đến ngày nhận lãi kế tiếp vào giá yết. Khoản lãi tích tụ này phải thuộc về người bán và được tính căn cứ vào lãi suất coupon điều chỉnh cho số ngày từ lần trả lãi trước cho đến ngày giao dịch trên số ngày trong kỳ trả lãi.

Số ngày từ lần trả lãi trước cho đến ngày giao dịch được tính bằng hàm COUPDAYBS với định dạng:

$$\text{COUPDAYBS}(\text{settlement}, \text{maturity}, \text{frequency}, \text{basis})$$

Các thông tin cần khai báo cho công thức trên gồm có ngày thanh toán (settlement), ngày đáo hạn (maturity), số lần trả lãi trong năm (frequency) và hệ đếm ngày (basis).

Ta nhập công thức sau vào ô C14 và tính được 90 ngày:

$$=\text{COUPDAYBS}(\text{C4}, \text{C5}, \text{C9}, \text{C10})$$

Để tính số ngày trong kỳ trả lãi, ta dùng hàm COUPDAYS với định dạng:

$$\text{COUPDAYS}(\text{settlement}, \text{maturity}, \text{frequency}, \text{basis})$$

Các thông tin cần khai báo vẫn là ngày thanh toán (settlement), ngày đáo hạn (maturity), số lần trả lãi trong năm (frequency) và hệ đếm ngày (basis).

Ta nhập công thức sau vào ô C15 và tính được 365 ngày:

$$=\text{COUPDAYS}(\text{C4}, \text{C5}, \text{C9}, \text{C10})$$

Lãi tích tụ bằng lãi suất tính cho năm, chia cho số lần trả lãi trong năm (để quy về lãi trong kỳ trả lãi), rồi nhân với số ngày tính lãi trên số ngày trong kỳ trả lãi. Trong ô C16, ta nhập công thức sau:

$$=(\text{C13}/\text{C14}) * \text{C5} * \text{C7}/\text{C8}$$

Lãi tích tụ tính được theo công thức trên bằng 2,219 đồng.

Trong ô C17, giá thanh toán của trái phiếu bằng giá sạch cộng với lãi tích tụ nên được tính theo công thức:


$$=\text{C13} + \text{C16}$$

Giá thanh toán tính được là 96,965. Với mệnh giá 100.000 đồng, giá thực tế thanh toán của trái phiếu là 96.965 đồng.

Mô hình vừa thiết lập ở trên cho ta kết quả định giá trái phiếu một cách tự động sau khi nhập các thông số đầu vào. Kết quả định giá theo mô hình Excel này hoàn toàn khớp với nguyên lý định giá trái phiếu theo ngân lưu chiết khấu.

Hình 3 minh họa sơ đồ ngân lưu trái phiếu. Sau khi mua trái phiếu vào ngày 16/02/2009, nếu giữ đến khi đáo hạn thì nhà đầu tư sẽ có 11 kỳ hưởng lãi vào các ngày 18/11 từ năm 2009 đến 2019. Tức là, $n = 11$. Vào mỗi kỳ trả lãi, khoản lãi nhà đầu tư nhận được là: $C = 9\% * 100 = 9$. Đến khi đáo hạn, nhà đầu tư nhận lại giá trị gốc là $F = 100$. Điểm phức tạp là trong khi các kỳ trả lãi cách đều nhau 1 năm thì thời gian từ ngày mua trái phiếu cho đến ngày trả lãi kế tiếp chỉ là $d = 275$ ngày. Đó là do trên thực tế ngày giao dịch hay ngày định giá nằm giữa 2 kỳ trả lãi, chứ không phải là vào đúng ngày trả lãi như thường được giả định trong những ví dụ đơn giản. Mô hình chiết khấu ngân lưu phải điều chỉnh cho kỳ hạn ngắn hơn 1 năm này.

Hình 3: Sơ đồ ngân lưu trái phiếu



Trước hết, ta định giá trái phiếu tại thời điểm của lần trả lãi kế tiếp, ngày 18/11/2009. Tức là, ta chiết khấu tất cả các khoản lãi và nợ gốc sẽ nhận được trong tương lai về thời điểm 18/11/2009.

$$P_1 = C + \frac{C}{1+y} + \frac{C}{(1+y)^2} + \dots + \frac{C}{(1+y)^{n-1}} + \frac{F}{(1+y)^{n-1}}$$

Chiết khấu về hiện tại trong khoảng thời gian d ngày, cho ta giá trị trái phiếu vào ngày 16/2/2009 như sau:

$$P = \frac{P_1}{(1+y)^{\frac{d}{365}}}$$

Đơn giản hơn, ta có thể sử dụng công thức dưới đây. Tuy nhiên, công thức này sẽ cho kết quả không hoàn toàn khớp với kết quả của mô hình Excel.

$$P = \frac{P_1}{1 + \frac{d}{365} y}$$

Vậy, TP4A4804 có giá vào ngày 16/2/2009 bằng:

$$\begin{aligned} P &= \frac{1}{(1+y)^{\frac{d}{365}}} \left[C + \frac{C}{1+y} + \frac{C}{(1+y)^2} + \dots + \frac{C+F}{(1+y)^{n-1}} \right] \\ &= \frac{1}{(1+9,8\%)^{\frac{275}{365}}} \left[9 + \frac{9}{1+9,8\%} + \frac{9}{(1+9,8\%)^2} + \dots + \frac{109}{(1+9,8\%)^{10}} \right] = 96,965 \end{aligned}$$

Tính lợi suất đến khi đáo hạn

Ở bảng tính “YTM”, nhập tựa đề “LỢI SUẤT ĐẾN KHI ĐÁO HẠN” vào ô B1 VÀ “NHẬP DỮ LIỆU” vào ô B3.

Nhập các tiêu đề: “Ngày thanh toán”, “Ngày đáo hạn”, “Lãi suất hàng năm”, “Giá yết (% mệnh giá)”, “Giá trị hoàn trả nợ gốc (% mệnh giá)”, “Số lần trả lãi trong năm” và “Hệ đếm ngày” lần lượt vào các ô B4 đến B10. Nhập thông tin C4-C6 và C8-C10 giống như các ô tương ứng trong bảng tính “Pricing”.

Tính lợi suất đến khi đáo hạn là bài toán ngược của định giá trái phiếu, trong đó đòi hỏi ta phải biết sử dụng giá trái phiếu làm thông tin đầu vào. Trong ô C7, nhập giá yết của trái phiếu là 94,746.

Hình 4: Tính lợi suất đến khi đáo hạn

	B	C	D	E
1	LỢI SUẤT ĐẾN KHI ĐÁO HẠN			
2				
3	NHẬP DỮ LIỆU			
4	Ngày thanh toán	16/02/2009	C4=DATE(2009,2,16)	
5	Ngày đáo hạn	18/11/2019	C5=DATE(2019,11,18)	
6	Lãi suất coupon hàng năm	9.00%		
7	Giá yết (% mệnh giá)	94.746		
8	Giá trị hoàn trả nợ gốc (% mệnh giá)	100		
9	Số lần trả lãi trong năm	1		
10	Hệ đếm ngày	1		
11				
12	KẾT QUẢ			
13	Số ngày kể từ lần trả lãi trước	90	C13=COUPDAYBS(C4,C5,C9,C10)	
14	Số ngày trong kỳ trả lãi	365	C14=COUPDAYS(C4,C5,C9,C10)	
15	Lãi tích tụ	2.219	C15=(C13/C14)*C6*C8/C9	
16	Giá thanh toán (% mệnh giá)	96.965	C16=C7+C15	
17	Lợi suất đến khi đáo hạn	9.800%	C17=YIELD(C4,C5,C6,C7,C8,C9,C10)	

Trong khung thông tin kết quả định giá, ta nhập tiêu đề “KẾT QUẢ” vào ô B12. Sau đó, nhập “Số ngày kể từ lần trả lãi trước”, “Số ngày trong kỳ trả lãi”, “Lãi tích tụ”, “Giá yết (% mệnh giá)” và “Lợi suất đến khi đáo hạn” lần lượt vào các ô B13 đến B17.

Nhập công thức tính ngày và giá thanh toán trong các ô C13-C16 như trong Hình 2 và tương tự như bảng tính “Pricing”.

Ta sử dụng hàm YIELD trong Excel để tính lợi suất đến khi đáo hạn của trái phiếu với định dạng:

YIELD(settlement,maturity,rate,pr,redemption,frequency,basis)

Các thông tin cần khai báo để định giá trái phiếu gồm có ngày thanh toán (settlement), ngày đáo hạn (maturity), lãi suất coupon (rate), giá yết (pr)⁴, giá trị hoàn trả nợ gốc (redemption), số lần trả lãi trong năm (frequency) và hệ đếm ngày (basis).

Dưới đây là công thức nhập trong ô C17 để tính lợi suất với kết quả là 9,8%:

⁴ Lưu ý rằng giá trái phiếu nhập trong hàm YIELD là giá yết chưa gộp lãi (giá sạch), chứ không phải là giá thanh toán thực tế.

$$=PRICE(C4,C5,C6,C7,C8,C9,C10)$$

Kết quả tính được từ mô hình Excel cũng khớp hoàn toàn với việc tính toán theo nguyên lý chiết khấu ngân lưu. Theo định nghĩa, lợi suất đến khi đáo hạn bằng suất chiết khấu ở đó giá trị hiện tại của lãi và vốn gốc hoàn trả trong ương lại của trái phiếu bằng đúng số tiền bỏ ra bây giờ.

$$P = \frac{1}{(1+y)^{\frac{d}{365}}} \left[C + \frac{C}{1+y} + \frac{C}{(1+y)^2} + \dots + \frac{C+F}{(1+y)^{n-1}} \right]$$

$$96,965 = \frac{1}{(1+y)^{\frac{275}{365}}} \left[9 + \frac{9}{1+y} + \frac{9}{(1+y)^2} + \dots + \frac{109}{(1+y)^{10}} \right]$$

Giải phương trình bằng phép thử hay dùng goal seek trong Excel, ta có:

$$YTM = y = 9,8\%$$